

# Active Server Pages (ASP)

## Outline

- 11.1 Introduction
- 11.2 How Active Server Pages Work
- 11.3 Client-side Scripting versus Server-side Scripting
- 11.4 Using Personal Web Server or Internet Information Server
- 11.5 A Simple ASP Example
- 11.6 Server-side ActiveX Components
- 11.7 File System Objects
- 11.8 Session Tracking and Cookies
- 11.9 Accessing a Database from an Active Server Page
- 11.10 Case Study: A Product Catalog

## 11.1 Introduction

- *Active Server Pages (ASP)*
  - Processed in response to client request
  - ASP file contains HTML and scripting code
  - VBScript de facto language for ASP scripting
    - Other languages can be used
      - JavaScript
  - **.asp** file extension
  - Microsoft-developed technology
  - Send dynamic Web content
    - HTML
    - DHTML
    - ActiveX controls
    - Client-side scripts
    - Java applets

## 11.2 How Active Server Pages Work

- Client sends request
  - Server receives request and directs it to ASP
  - ASP processes, then returns result to client
- HTTP request types
  - Request methods
    - **GET**
      - Gets (retrieves) information from server
      - Retrieve HTML document or image
    - **POST**
      - Posts (sends) data to server
      - Send info from HTML form
        - » Client-entered data
        - » Info to search Internet
        - » Query for a database
        - » Authentication info

## 11.2 How Active Server Pages Work (II)

- Browsers often cache Web pages
  - *Cache*: save on disk
  - Typically do not cache **POST** response
    - Next **POST** request may not return same result
- Client requests ASP file
  - Parsed (top to bottom) by ActiveX component **asp.dll**
    - *ActiveX component*: server-side ActiveX control that usually does not have GUI
    - Code executed as encountered
    - **@LANGUAGE** statement
      - Specifies scripting language
      - If not used, VBScript assumed
    - As interpreted, HTML (plus client-side scripts) sent to client
  - Parsed each time requested
  - Web server must support ASP by providing component such as **asp.dll**

## 11.3 Client-side Scripting versus Server-side Scripting

- Client-side scripting
  - Used for:
    - Validation
    - Interactivity
    - Enhancing Web page with ActiveX controls
    - Dynamic HTML
    - Java applets
    - Accessing browser
  - Browser-dependent
    - Scripting language must be supported by browser or scripting host
  - Viewable on client
    - Protecting source code difficult

## 11.3 Client-side Scripting versus Server-side Scripting (II)

- Server-side scripting
  - Reside on server → more flexibility
    - Database access
  - Usually generate custom response for client
  - Access to ActiveX server components
    - Extend scripting language functionality
  - Run exclusively on server → cross-platform issues not a concern
  - Not visible to client
    - Only HTML + client-side scripts sent to client

## 11.4 Using Personal Web Server or Internet Information Server

- Create subdirectories in root
  - **Deitel\_Iw3http**
    - Copy all **.asp** files
  - **includes**
    - Copy all **.inc** files
  - **images**
    - Copy all **.gif** files
  - May have to change paths in **.asp** files to reflect directories

## 11.5 A Simple ASP Example

- Scripting delimiters
  - `<% and %>`
  - Indicate code is to be executed on server, not client
- **@LANGUAGE**
  - Specify scripting language (default VBScript)
  - `<% @LANGUAGE = "VBScript" %>`
- Each time page refreshed, server loads and interprets ASP



```

1<% @LANGUAGE = VBScript %>
2<% Option Explicit %>
3<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.0 Transitional//EN">
4<% ' Fig. 11.1 : clock.asp %>
5
6<HTML>
7<HEAD>
8<TITLE>A Simple ASP Example</TITLE>
9<META HTTP-EQUIV = "REFRESH" CONTENT = "60; URL = CLOCK.ASP">
10</HEAD>
11<BODY>
12
13<FONT FACE = ARIAL SIZE = 4><STRONG>Simple ASP Example</STRONG>
14</FONT><P>
15   <TABLE BORDER = "6">
16   <TR>
17   <TD BGCOLOR = "#000000">
18       <FONT FACE = Arial COLOR = "#00FF00" SIZE = 4>
19           <% =Time() %>
20       </FONT>
21   </TD>
22   </TR>
23   </TABLE>
24</BODY>
25</HTML>

```

**1.1 Specify VBScript as scripting language**



**1.2 Use Option Explicit to indicate variables be explicitly declared by programmer**

**1.3 Use META tag to set refresh interval**

**Time gets current time on server (hh:mm:ss)**

**Short for <% Call Response.Write( Time() ) %>**

# Output from a simple Active Server Page



## 11.6 Server-side ActiveX Components

- Server-side ActiveX components
  - Typically do not have GUI
  - If scripting language for ASP not support certain feature, create ActiveX Server component
    - Visual C++, Visual Basic, Delphi, etc.
  - Usually execute faster than scripting language equivalents
  - Executed on server
    - Client does not need to support ActiveX technologies

## Some server-side ActiveX components included with IIS and PWS

Component Name	Description
<b>MSWC.BrowserType</b>	ActiveX component for gathering information (e.g., type, version, etc.) about the client's browser.
<b>MSWC.AdRotator</b>	ActiveX component for rotating advertisements on a Web Page.
<b>MSWC.NextLink</b>	ActiveX component for linking together Web pages.
<b>MSWC.ContentRotator</b>	ActiveX component for rotating HTML content on a Web page.
<b>MSWC.PageCounter</b>	ActiveX component for storing the number of times a Web page has been requested.
<b>MSWC.Counters</b>	ActiveX components that provides general-purpose persistent counters.
<b>MSWC.MyInfo</b>	ActiveX component that provides information (e.g., owner name, owner address, etc.) about a Web site.
<b>Scripting.FileSystemObject</b>	ActiveX component that provide an object library for accessing files on the server or on the server's network.
ActiveX Data Objects (ADO) Data Access Components	ActiveX components that provide an object library for accessing databases.

```

1<% @LANGUAGE = VBScript %>
2<% Option Explicit %>
3<% ' Fig. 11.3 : rotate.asp %>
4
5<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.0 Transitional//EN">
6<HTML>
7<HEAD>
8<TITLE>AdRotator Example</TITLE>
9</HEAD>
10
11<BODY>
12<FONT FACE = "Arial" SIZE = 4><STRONG>AdRotator Example</STRONG>
13</FONT><P>
14<%
15  ' Declare flagChanger
16  Dim flagChanger
17
18  ' Create an AdRotator object
19  Set flagChanger = Server.CreateObject( "MSWC.AdRotator" )
20
21  ' Use config.txt to send an advertisement to the client
22  Call Response.Write( _
23      flagChanger.GetAdvertisement( "config.txt" ) )
24%>
25</BODY>
26</HTML>

```

**1.1 Create instance of AdRotator component**

**1.2 Call Response object's Write method to send advertisement to client**

```
27 WIDTH 54
28 HEIGHT 36
29 BORDER 1
30 *
31 /images/us.gif
32 http://www.odci.gov/cia/publications/factbook/us.html
33 United States Information
34 20
35 /images/france.gif
36 http://www.odci.gov/cia/publications/factbook/fr.html
37 France Information
38 20
39 /images/germany.gif
40 http://www.odci.gov/cia/publications/factbook/gm.html
41 Germany Information
42 20
43 /images/italy.gif
44 http://www.odci.gov/cia/publications/factbook/it.html
45 Italy Information
46 20
47 /images/spain.gif
48 http://www.odci.gov/cia/publications/factbook/sp.html
49 Spain Information
50 20
```

config.txt

1. Header includes image HEIGHT, image WIDTH and image BORDER width

Asterisk separates header from advertisements




Image location

Destination URL

ALT tag

Percentage of time image appears

# Demonstrating the AdRotator ActiveX component



## 11.7 File System Objects

- File System Objects (FSOs)
  - Manipulate files, directories and drives
  - Read and write text
  - In Microsoft Scripting Runtime Library
  - 5 FSO types:
    - **FileSystemObject**
      - Interact with **Files**, **Folders** and **Drives**
    - **File**
      - Manipulate **Files** of any type
    - **Folder**
      - Manipulate **Folders** (i.e, directories)
    - **Drive**
      - Gather info about **Drives** (local or remote)
    - **TextStream**
      - Read and write text files



## 11.7 File System Objects

- `OpenTextFile(filename, code, create)`
  - `filename` - file to open
  - `code`
    - 8 - open for appending
    - 1 - open for reading
    - 2 - open for writing
  - `create?` - if `true`, creates a new file if does not exist

# FileSystemObject methods

Methods	Description
<b>CopyFile</b>	Copies an existing <b>File</b> .
<b>CopyFolder</b>	Copies an existing <b>Folder</b> .
<b>CreateFolder</b>	Creates and returns a <b>Folder</b> .
<b>CreateTextFile</b>	Creates and returns a text <b>File</b> .
<b>DeleteFile</b>	Deletes a <b>File</b> .
<b>DeleteFolder</b>	Deletes a <b>Folder</b> .
<b>DriveExists</b>	Tests whether or not a <b>Drive</b> exists.
<b>FileExists</b>	Tests whether or not a <b>File</b> exists. Returns boolean.
<b>FolderExists</b>	Tests whether or not a <b>Folder</b> exists. Returns boolean.
<b>GetAbsolutePathName</b>	Returns the absolute path as a string.
<b>GetDrive</b>	Returns the specified <b>Drive</b> .
<b>GetDriveName</b>	Returns the <b>Drive</b> drive name.
<b>GetFile</b>	Returns the specified <b>File</b> .
<b>GetFileName</b>	Returns the <b>File</b> file name.
<b>GetFolder</b>	Returns the specified <b>Folder</b> .
<b>GetParentFolderName</b>	Returns a string representing the parent folder name.
<b>GetTempName</b>	Creates and returns a string representing a file name.
<b>MoveFile</b>	Moves a <b>File</b> .
<b>MoveFolder</b>	Moves a <b>Folder</b> .
<b>OpenTextFile</b>	Opens an existing text <b>File</b> . Returns a <b>TextStream</b> .

# Some common File properties and methods

Property/method	Description
<i>Properties</i>	
<b>DateCreated</b>	Date. The date the <b>File</b> was created.
<b>DateLastAccessed</b>	Date. The date the <b>File</b> was last accessed.
<b>DateLastModified</b>	Date. The date the <b>File</b> was last modified.
<b>Drive</b>	Drive. The <b>Drive</b> where the file is located.
<b>Name</b>	String. The <b>File</b> name.
<b>ParentFolder</b>	String. The <b>File</b> 's parent folder name.
<b>Path</b>	String. The <b>File</b> 's path.
<b>ShortName</b>	String. The <b>File</b> 's name expressed as a short name.
<b>Size</b>	The size of the <b>File</b> in bytes.
<i>Methods</i>	
<b>Copy</b>	Copy the <b>File</b> . Same as <b>CopyFile</b> of <b>FileSystemObject</b> .
<b>Delete</b>	Delete the <b>File</b> . Same as <b>DeleteFile</b> of <b>FileSystemObject</b> .
<b>Move</b>	Move the <b>File</b> . Same as <b>MoveFile</b> of <b>FileSystemObject</b> .
<b>OpenAsTextStream</b>	Opens an existing <b>File</b> as a text <b>File</b> . Returns <b>TextStream</b> .

## Some Folder properties and methods

Property/method	Description
<i>Properties</i>	
<b>Attributes</b>	Integer. Value indicating <b>Folder</b> 's attributes (read only, hidden, etc.)
<b>DateCreated</b>	Date. The date the folder was created.
<b>DateLastAccessed</b>	Date. The date the folder was last accessed.
<b>DateLastModified</b>	Date. The date the folder was last modified.
<b>Drive</b>	Drive. The <b>Drive</b> where the folder is located.
<b>IsRootFolder</b>	Boolean. Indicates whether or not a <b>Folder</b> is the root folder.
<b>Name</b>	String. The <b>Folder</b> 's name.
<b>ParentFolder</b>	String. The <b>Folder</b> 's parent folder name.
<b>Path</b>	String. The <b>Folder</b> 's path.
<b>ShortName</b>	String. The <b>Folder</b> 's name expressed as a short name.
<b>ShortPath</b>	String. The <b>Folder</b> 's path expressed as a short path.
<b>Size</b>	Variant. The total size in bytes of all subfolders and files.
<b>Type</b>	String. The <b>Folder</b> type.
<i>Methods</i>	
<b>Delete</b>	Delete the <b>Folder</b> . Same as <b>DeleteFolder</b> of <b>FileSystemObject</b> .
<b>Move</b>	Move the <b>Folder</b> . Same as <b>MoveFolder</b> of <b>FileSystemObject</b> .
<b>Copy</b>	Copy the <b>Folder</b> . Same as <b>CopyFolder</b> of <b>FileSystemObject</b> .

# Drive properties

Property	Description
<b>AvailableSpace</b>	Variant. The amount of available <b>Drive</b> space in bytes.
<b>DriveLetter</b>	String. The letter assigned the <b>Drive</b> (e.g., "C").
<b>DriveType</b>	Integer. The <b>Drive</b> type. Constants <b>Unknown</b> , <b>Removable</b> , <b>Fixed</b> , <b>Remote</b> , <b>CDRom</b> and <b>RamDisk</b> represent <b>Drive</b> types and have the values 0–5, respectively.
<b>FileSystem</b>	String. The file system <b>Drive</b> description (FAT, FAT32, NTFS, etc.).
<b>FreeSpace</b>	Variant. Same as <b>AvailableSpace</b> .
<b>IsReady</b>	Boolean. Indicates whether or not a <b>Drive</b> is ready for use.
<b>Path</b>	String. The <b>Drive</b> 's path.
<b>RootFolder</b>	Folder. The <b>Drive</b> 's root <b>Folder</b> .
<b>SerialNumber</b>	Long. The <b>Drive</b> serial number.
<b>TotalSize</b>	Variant. The total <b>Drive</b> size in bytes.
<b>VolumeName</b>	String. The <b>Drive</b> volume name.

```

1  <% @LANGUAGE = VBScript %>
2  <% Option Explicit %>
3
4  <% ' Fig. 11.10 : guestbook.asp %>
5  <%
6      Dim fso, tso, guestBook, mailtoUrl
7
8      If Request( "entry" ) = "true" Then
9
10         ' Instantiate a FileSystem Object
11         Set fso = CreateObject( "Scripting.FileSystemObject" )
12
13         ' Guestbook must be open for write by all!
14         ' guestbook path must be modified to reflect the file
15         ' structure of the server
16         guestBook = "c:\webshare\wwwroot\deitel_iw3http\" & _
17                 "guestbook.txt"
18
19         ' Check if the file exists, if not create it
20         If fso.FileExists( guestbook ) <> True Then
21             Call fso.CreateTextFile( guestBook )
22         End If
23
24         ' Open the guestbook and create the entry
25         Set tso = fso.OpenTextFile( guestbook, 8, True )
26
27         ' Build the mailtoUrl
28         mailtoUrl = Request( "pubdate" ) & " <A HREF=" & Chr( 34 )
29         mailtoUrl = mailtoUrl & "mailto:" & Request( "email" )
30         mailtoUrl = mailtoUrl & Chr( 34 ) & ">" & Request( "name" )
31         mailtoUrl = mailtoUrl & "</A>:  "
32

```

**1.1 Test if entry variable "true"; when page first requested has value ""; passed value "true" during POST operation**

**1.2 Create FSO instance**

```

33      ' Send the information to guestbook.txt
34      Call tso.WriteLine( "<HR COLOR = BLUE SIZE = 1>" )
35      Call tso.WriteLine( mailtoUrl )
36      Call tso.WriteLine( Request( "comment" ) )
37      Call tso.Close()
38  End If
39%>
40<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.0 Transitional//EN">
41<HTML>
42<HEAD>
43<TITLE> GuestBook Example</TITLE>
44
45<BODY>
46<FONT SIZE = 2 FACE = "arial" COLOR = "blue">
47
48
49<%
50      ' Print a thank you if they have made a request
51      If Request( "entry" ) = "true" Then
52%>
53
54Thanks for your entry!<P>
55<% Else %>
56Please leave a message in our guestbook.<P>
57<% End If %>
58
59</FONT>
60<FORM ACTION = "guestbook.asp?entry=true" METHOD = "POST">
61
62<CENTER>
63<INPUT TYPE = "hidden" NAME = "pubdate" VALUE = "<% =Date() %>">
64

```

### 1.3 Specify POST operation

### 1.4 Create variable entry with value true

```

65<TABLE>
66<TR>
67<TD><FONT FACE = "arial" SIZE = "2">Your Name:  </FONT><BR></TD>
68<TD><INPUT TYPE = "text" FACE = ARIAL
69     SIZE = "60" NAME = "name"><BR></TD>
70</TR>
71<TR>
72<TD><FONT FACE = "arial" SIZE = "2">Your email address:
73</FONT><BR></TD>
74<TD><INPUT TYPE = "text" FACE = "arial" SIZE = "60" NAME = "email"
75VALUE = "user@isp.com"><BR></TD>
76</TR>
77<TR>
78<TD><FONT FACE = "arial" SIZE = "2">Tell the world:  </FONT>
79<BR></TD>
80<TD><TEXTAREA NAME = "comment" ROWS = "3" COLS = "50">
81Replace this text with the information
82you would like to post.
83</TEXTAREA><BR></TD>
84</TR>
85</TABLE>
86<INPUT TYPE = "submit" VALUE = "SUBMIT">
87<INPUT TYPE = "reset" VALUE = "CLEAR">
88</CENTER>
89</FORM>
90
91<%
92     Dim fso2, tso2
93
94     ' Instantiate a FileSystem Object
95     Set fso2 = Server.CreateObject( _


```

## 1.5 Display contents of guestbook.txt



```
96     "Scripting.FileSystemObject" )
97
98     ' Guestbook must be open for write by all!
99     ' Guestbook path must be modified to reflect
100    ' the file structure of the server
101    guestBook = "c:\webshare\wwwroot\deitel_iw3http\" & _
102              "guestbook.txt"
103
104    ' Check if the file exists, if not create it
105    If fso2.FileExists( guestBook ) = True Then
106
107        ' Open the guestbook, "1" is for reading
108        Set tso2 = fso2.OpenTextFile( guestbook, 1 )
109
110        ' Read the entries
111        Call Response.Write( "Guestbook Entries:<BR>" )
112        Call Response.Write( tso2.ReadAll() )
113        Call tso2.Close()
114    End If
115%>
116
117</BODY>
118
119</HTML>
```

# Guest book Active Server Page



GuestBook Example - Microsoft Internet Explorer

File Edit View Favorites Tools Help

Address <http://localhost/guestbook.asp>


Please leave a message in our guestbook.

Your Name:

Your email address:

Tell the world:

Done Local intranet



GuestBook Example - Microsoft Internet Explorer

File Edit View Favorites Tools Help

Address <http://localhost/guestbook.asp>


Please leave a message in our guestbook.

Your Name:

Your email address:

Tell the world:

Done Local intranet



GuestBook Example - Microsoft Internet Explorer

File Edit View Favorites Tools Help

Address <http://localhost/guestbook.asp?entry=true>

Thanks for your entry!

Your Name:

Your email address:

Tell the world:

Guestbook Entries:

---

9/19/99 [Test User](#): This page works great!

Done Local intranet

## 11.8 Session Tracking and Cookies

- Customized services
  - <http://my.yahoo.com/>
  - Shopping cart
  - Marketing
- Track clients
  - *Session tracking*
    - Handled by server
    - ASP: **Session** object
      - **Timeout** property
        - » Default: 20 minutes
        - » Terminate with **Abandon** method

## 11.8 Session Tracking and Cookies (II)

- *Cookies*
  - Small files sent as part of response to client
  - Every HTTP interaction between client and server includes *header*
    - Contains info about request or response
      - Request: type of request (**GET** or **POST**) and cookies stored on client machine
  - Maximum age
    - Duration of browsing session
    - Store cookies on client computer for future use
    - Cookies automatically deleted when expire

## 11.8 Session Tracking and Cookies (III)

- *ASP application*
  - Multiple ASP pages linked through **POST** requests
- Server side include (SSI)
  - SSI statement replaced with contents of file
  - SSI statements always execute before any scripting code
  - **.inc** extension
- Virtual paths vs. physical paths

```

1<% @LANGUAGE = VBScript %>
2<% Option Explicit %>
3
4<% ' Fig. 11.11 : instantpage.asp %>
5
6<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.0 Transitional//EN">
7<HTML>
8<HEAD>
9<TITLE>Instant Page Content Builder</TITLE>
10</HEAD>
11
12<BODY>
13
14<!-- include header goes here-->
15<!-- #include virtual="/includes/mgtheadr.inc" -->
16
17<FONT FACE = "ARIAL" SIZE = "2">
18<H2>Instant Page Content Builder</H2>
19
20<%
21  ' If there is a value from process.asp, show it here
22  Call Response.Write( Session( "fileNameString" ) & "<BR>" )
23
24  ' Begin the form
25%>
26<FORM ACTION = "process.asp" METHOD = "POST">
27
28<CENTER>
29<INPUT TYPE = "hidden" NAME = "pubdate" VALUE = "<% =Date() %>">
30<TABLE>
31<TR>

```

**1.1 Use SSI statement to include mgtheadr.inc**

**1.2 Retrieve value of session variable fileNameString**

**1.3 Load process.asp in response to POST request**

```
32<TD><FONT FACE = "arial" SIZE = "2">Your Name:
33</FONT><BR></TD>
34<TD><INPUT TYPE = "text" FACE = "arial" SIZE = "60"
35     NAME = "name"><BR></TD>
36</TR>
37<TR>
38<TD><FONT FACE = "arial" SIZE = "2"> Enter the Filename:
39</FONT><BR></TD>
40<TD><INPUT TYPE = "text" FACE = "arial" SIZE = "60"
41     NAME = "filename" VALUE = "YourFileName.asp"><BR></TD>
42</TR>
43<TR>
44<TD><FONT FACE = "arial" SIZE = "2">Enter the Title:
45</FONT><BR></TD>
46<TD><INPUT TYPE = "text" FACE = "arial" SIZE = "60"
47     NAME = "doctitle" VALUE = "Document Title"><BR></TD>
48</TR>
49<TR>
50<TD><FONT FACE = "arial" SIZE = "2">Enter the Content:
51</FONT><BR></TD>
52<TD><TEXTAREA NAME = "content" ROWS = "3" COLS = "50">
53Replace this text with the
54information you would like to post.
55</TEXTAREA><BR></TD>
56</TR>
57</TABLE>
58<INPUT TYPE = "submit" VALUE = "SUBMIT">
59<INPUT TYPE = "reset" VALUE = "CLEAR">
60</CENTER>
61</FORM>
62
```

```
63 <!-- #include virtual="/includes/mgtfooter.inc" -->
64
65 </BODY>
66 </HTML>
```

**1.4 Use SSI statement  
to include  
mgtfooter.inc**

```
67<HR SIZE = 1 COLOR = BLUE>
68<IMG HEIGHT = 100 SRC = "/images/mgheader.gif">
69<HR SIZE = 1 COLOR = BLUE>
```


**mgtheadr.inc**

```
70<HR COLOR = BLUE SIZE = 1>
71<CENTER>
72<A HREF = "mailto:orders">Ordering Information</A> -
73<A HREF = "mailto:editor">Contact the Editor</A><BR>
74<HR COLOR = BLUE SIZE = 1>
75</CENTER>
76</FONT>
```

**mgtfooter.inc**



# Creating Web pages using an Active Server



The screenshot shows a Microsoft Internet Explorer browser window titled "Instant Page Content Builder - Microsoft Internet Explorer". The address bar displays "http://localhost/instantpage.asp". The page content includes a logo for "Mother Goose Translations" featuring a yellow star and a blue goose. Below the logo is the heading "Instant Page Content Builder". The form contains four input fields: "Your Name:" (empty), "Enter the Filename:" (containing "YourFileName.asp"), "Enter the Title:" (containing "Document Title"), and "Enter the Content:" (containing "Replace this text with the information you would like to post."). There are "SUBMIT" and "CLEAR" buttons below the content field. At the bottom, there is a link for "Ordering Information - Contact the Editor". The status bar at the bottom shows "Done" and "Local intranet".


Instant Page Content Builder - Microsoft Internet Explorer

File Edit View Favorites Tools Help

Back Forward Stop Refresh Home Search Favorites History Mail Print

Address <http://localhost/instantpage.asp> Go Links

---



---

## Instant Page Content Builder

Your Name:

Enter the Filename:

Enter the Title:

Enter the Content:

---

[Ordering Information - Contact the Editor](#)

---

Done Local intranet

```

77 <% @LANGUAGE = VBScript %>
78 <% Option Explicit %>
79
80 <% ' Fig. 11.14 : process.asp %>
81 <%
82     Dim header, footer, directoryPath, filePathname, fileName
83     Dim fileNameString, newTitle, fso, tso
84
85     ' Build the header
86     ' vbCrLf inserts a carriage return/linefeed into the text
87     ' string which makes the HTML code more legible
88     header = "<HTML>" & vbCrLf & "<HEAD>" & vbCrLf _
89         & "<META NAME= " & Chr( 34 ) & "author" & Chr( 34 ) _
90         & " CONTENT = " & Chr( 34 ) & Request( "name" ) _
91         & Chr( 34 ) & ">" & vbCrLf & "<META NAME= " & Chr( 34 ) _
92         & "pubdate" & Chr( 34 ) & " CONTENT = " & Chr( 34 ) _
93         & request( "pubdate" ) & Chr( 34 ) & ">" & vbCrLf _
94         & "<TITLE>" & request( "doctitle" ) & "</TITLE>" _
95         & vbCrLf & "</HEAD>" & vbCrLf & "<BODY>" _
96         & vbCrLf & "<FONT FACE=" & Chr( 34 ) & "arial" _
97         & Chr( 34 ) & " SIZE=2>" & vbCrLf _
98         & "<!-- #include virtual=" & Chr( 34 ) _
99         & "/includes/mgtheader.inc" & Chr( 34 ) & " -->" _
100        & vbCrLf & "<H1>" & Request("doctitle") & "</H1>" & vbCrLf
101
102     ' Build the footer using a different method for
103     ' building the string
104     footer = "<!-- #include virtual=" & Chr( 34 )
105     footer = footer & "/includes/mgtfooter.inc"
106     footer = footer & Chr( 34 ) & " -->" & vbCrLf
107     footer = footer & "</FONT>" & vbCrLf
108     footer = footer & "</BODY>" & vbCrLf

```

**1.1 Build header and assign HTML to string header**

**1.2 Build footer using different method and assign HTML to string footer**

```

109 footer = footer & "</HTML>" & vbCrLf
110
111 ' Creates a FileSystem Object
112 Set fso = Server.CreateObject( "Scripting.FileSystemObject" )
113
114 ' Directory must be open for write by all!
115 ' directoryPath must be modified to reflect the file structure
116 ' of the server
117 directoryPath = "c:\webshare\wwwroot\temp\"
118
119 'check to make sure that the filename is not empty
120 If Request( "filename" ) = "YourFileName.asp" Then
121     Session( "filenamestring" ) = _
122         "<FONT COLOR=" & Chr( 34 ) & "red" _
123         & Chr( 34 ) & "SIZE=4>" _
124         & "Please enter a unique filename.</FONT>"
125     Call Response.Redirect( "instantpage.asp" )
126 End If
127
128 ' Builds path for text file
129 filePathname = directoryPath & Request( "filename" )
130
131 ' See if the directory exists
132 If fso.FolderExists( directoryPath ) <> True Then
133     Call fso.CreateFolder( directoryPath )
134 End If
135
136 ' Check if the filename already exists
137 If fso.FileExists( filePathname ) = True Then
138     Session( "filenamestring" ) = "<FONT COLOR=" _
139         & Chr( 34 ) & "red" & Chr( 34 ) & "SIZE=" _
140         & Chr( 34 ) & "4" & Chr( 34 ) & ">" _

```

**1.3 Create FSO object**

**1.4 Validate contents of text box “Enter the Filename”**

**1.5 Use Response method Redirect to redirect browser**

**1.6 Preserve state with session variable filenamestring**

**1.7 Check to see if file name is unique**

```

141         & "That filename is being used." _
142         & "Please enter a unique filename.</FONT>"
143     Call Response.Redirect( "instantpage.asp" )
144 End If
145
146 ' If fileNameString does not get set by an error,
147 ' give it a friendly value so that
148 ' when the user repeats the process,
149 ' fileNameString is initialized
150 Session( "filenamestring" ) = "<FONT COLOR=" _
151     & Chr( 34 ) & "blue" & Chr( 34 ) & "SIZE =" _
152     & Chr( 34 ) & "4" & Chr( 34 ) & ">" _
153     & "Welcome Back " & Request( "name" ) & "!</FONT>"
154
155 ' Create the html file
156 Set tso = fso.CreateTextFile( filePathName, False )
157 Call tso.WriteLine( header )
158 Call tso.WriteLine( Request( "content" ) )
159 Call tso.WriteLine( footer )
160 Call tso.Close
161
162 ' Create the results page
163 %>
164 <!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.0 Transitional//EN">
165 <HTML>
166 <HEAD>
167
168 <% ' Use the title given by the user %>
169 <TITLE>File Generated: <% =Request( "doctitle" ) %></TITLE>
170 </HEAD>
171


```

**1.8** If file name is unique, give filenamestring session variable a value indicating so

**1.9** Create the html document by writing the header, content and footer to the specified file

```
172<BODY>
173<!-- #include virtual="/includes/mgtheadheader.inc" -->
174<FONT FACE = "arial" SIZE = "2">
175<H1>File Generated: <% =Request( "doctitle" ) %></H1>
176
177<% ' Provide a link to the generated page %>
178Your file is ready:
179<A HREF = "/temp/<% =Request( "filename" ) %>">
180<% =Request( "doctitle" ) %></A>
181<P>
182<!-- #include virtual = "/includes/mgtfooter.inc" -->
183</FONT>
184</BODY>
185</HTML>
```


# ASP file that creates an ASP file for the user



```
1<HTML>
2<HEAD>
3<META NAME = "author" CONTENT = "J Smith">
4<META NAME = "pubdate" CONTENT = "9/19/99">
5<TITLE>Important Announcement</TITLE>
6</HEAD>
7<BODY>
8<FONT FACE = "arial" SIZE = 2>
9<!-- #include virtual = "/includes/mgtheader.inc" -->
10<H1>Important Announcement</H1>
11
12Instant Page Builder is ready!
13<!-- #include virtual = "/includes/mgtfooter.inc" -->
14</FONT>
15</BODY>
16</HTML>
```

**Sample page  
generated by  
instantpage.asp**

# Sample page generated by instantpage.asp





## 11.9 Accessing a Database from an Active Server Page

- ASP can communicate with databases
  - Via ADO (ActiveX Data Objects)
- *Three-tier distributed applications*
  - User interface
    - Often created using HTML, Dynamic HTML or XML
  - Business logic
    - Use Web servers
  - Database access
  - All three tiers may reside on separate computers connected to a network

```

1  <% @LANGUAGE = VBScript %>
2  <% Option Explicit %>
3
4  <% ' Fig. 11.16 : login.asp %>
5
6  <%
7      ' Set up the variables for this page
8      Dim dbConn, dbQuery, loginRS, loginFound
9
10     ' Check to see if there is an existing connection to
11     ' the Database. If not, create one
12     If IsObject( Session( "mothergoose_dbConn" ) ) Then
13         Set dbConn = Session( "mothergoose_dbConn" )
14     Else
15         Set dbConn = Server.CreateObject( "ADODB.Connection" )
16         Call dbConn.Open( "mothergoose", "", "" )
17         Set Session( "mothergoose_dbConn" ) = dbConn
18     End If
19
20     ' Create the SQL query
21     dbQuery = "SELECT * FROM users"
22
23     ' Create the recordset
24     Set loginRS = Server.CreateObject( "ADODB.Recordset" )
25     Call loginRS.Open( dbQuery, dbConn )
26
27     On Error Resume Next ' If an error occurs, ignore it
28
29     ' Move to the first record in the recordset
30     Call loginRS.MoveFirst()
31 %>
32

```

## 1.1 Test if session variable

mothergoose\_dbConn  
is storing an object

## 1.2 If not, create an ADODB.Connection object to open a connection

## 1.3 Create SQL query

## 1.4 Create recordset

```

33<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.0 Transitional//EN">
34<HTML>
35<HEAD><TITLE>Login Page</TITLE></HEAD>
36
37<BODY>
38<!-- include header goes here-->
39<!-- #include virtual = "/includes/mgtheadr.inc" -->
40
41<%
42  ' If this is a return after a failed attempt, print an error
43  If Session( "loginFailure" ) = True Then %>
44<FONT SIZE = 4 COLOR = "red"> Login attempt failed,
45please try again <P></FONT>
46  <% End If %>
47
48<% ' Begin the form %>
49<FONT FACE = "arial" SIZE = 2>
50Please select your name and enter
51your password to login:<BR>
52</FONT>
53<FORM NAME = sublogform ACTION = "submitlogin.asp" METHOD = POST>
54
55<% ' Format the form using a table %>
56<TABLE BORDER = 0>
57<TR>
58<TD><FONT FACE = "arial" SIZE = 2>Name:</FONT></TD>
59<TD><SELECT NAME = "LOGINID">
60<OPTION VALUE = "000">Select your name
61<%
62  ' Pull user names from the query to populate the dropdown
63  While Not loginRS.EOF
64

```

**1.5 Test if session variable loginFailure is True**

**1.6 Create dropdown menu from recordset**

```


65      ' If there is a session loginid, reuse it
66      If Session( "loginid" ) = loginRS( "loginid" ) Then
67          loginFound = "selected "
68      End If
69
70      ' If a login cookie was found, reuse it
71      If Request.Cookies( "loginid" ) = loginRS( "loginid" ) Then
72          loginfound = "selected "
73      End If
74
75      ' Create each dropdown entry %>
76<OPTION <% =loginFound %>
77VALUE = "<% =loginRS( "loginid" ) %>">
78<% =loginRS( "loginid" ) %>
79<% loginfound = " " %>
80<%
81      Call loginRS.MoveNext()
82  Wend
83%>
84</SELECT>
85</TD>
86</TR>
87
88<TR>
89<TD><FONT FACE = "arial" SIZE = "2">Password:</FONT></TD>
90<TD><INPUT TYPE = "password" NAME = "SUBMIT_LOGIN"></TD>
91</TR>
92<TR>
93<TD>&nbsp;</TD>
94<TD ALIGN = "LEFT"><INPUT TYPE = "submit" VALUE = "Log Me In"
95ID = "login1" NAME = "login1"></TD>
96</TR>

```

## 1.7 Test for session loginid and loginid cookie (written by submitlogin.asp)

```
97 </TABLE>
98 </FORM>
99 </FONT>
100
101<!-- #include virtual = "/includes/mgtfooter.inc" -->
102
103</BODY>
104
105</HTML>
```

## 2. Page rendered by browser



```

106 <% @LANGUAGE = VBScript %>
107 <% Option Explicit %>
108
109 <% ' Fig. 11.17 : submitlogin.asp %>
110 <%
111 ' Set up the variables for this page
112 Dim dbConn, dbQuery, loginRS
113
114 ' Check to see if there is an existing connection to
115 ' the Database. If not, create one
116 If IsObject( Session( "mothergoose_dbConn" ) ) Then
117     Set dbConn = Session( "mothergoose_dbConn" )
118 Else
119     Set dbConn = Server.CreateObject( "ADODB.Connection" )
120     Call dbConn.Open( "mothergoose", "", "" )
121     Set Session( "mothergoose_dbConn" ) = dbConn
122 End If
123
124 ' Create the SQL query
125 dbQuery = "SELECT * FROM users"
126
127 ' Create the recordset
128 Set loginRS = Server.CreateObject( "ADODB.Recordset" )
129 Call loginRS.Open( dbQuery, dbConn )
130
131 On Error Resume Next ' If an error occurs, ignore it
132
133 ' Move to the first record in the recordset
134 Call loginRS.MoveFirst()
135
136 ' If the loginid is not empty then
137 If Request( "loginid" ) <> "" Then

```

## 1.1 Check for database connection

## 1.2 Create recordset

```

138 While Not loginRS.EOF
139     If Request( "loginid" ) = loginRS( "loginid" ) AND _
140         Request( "submit_login" ) = loginRS( "password" ) Then
141
142         ' Password and loginid are OK set a Session variable
143         Session( "loginfailure" ) = False
144
145         ' Set a cookie to recognize them the next time they
146         ' go to login.asp
147         Response.Cookies( "loginid" ) = Request( "loginid" )
148
149         ' Send them on to the next page
150         Call Response.Redirect( "instantpage.asp" )
151     End If
152
153     Call loginRS.MoveNext()      ' Move on to the next record
154 Wend
155
156
157 ' If loginid is empty, or no match was found
158 Session( "loginFailure" ) = True ' Set loginFailure to true
159
160 ' Return to the login page
161 Call Response.Redirect( "login.asp" )
162%>

```

**1.3 Check to see if password and loginid match a record in recordset**

**1.4 If so, set session variable loginFailure to False and set a cookie for future use**

**1.5 If no match was found, set session variable loginFailure to True**

**1.6 Redirect browser to login.asp**

```

1<% @LANGUAGE = VBScript %>
2<% Option Explicit %>
3
4<% ' Fig. 11.18 : translation.asp %>
5
6<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.0 Transitional//EN">
7<HTML>
8<HEAD>
9<META HTTP-EQUIV = "CONTENT-TYPE"
10     CONTENT = "TEXT/HTML;CHARSET = WINDOWS-1252">
11<TITLE>Mother Goose Translations</TITLE>
12</HEAD>
13
14<BODY>
15<% ' Use the mothergoose header %>
16<!--#include virtual = "/includes/mgtheadheader.inc"-->
17
18<%
19  Dim conn, sql, rs, titleTest, image
20
21  ' Check if there is a mothergoose database connection open
22  If IsObject( Session( "mothergoose_conn" ) ) Then
23      Set conn = Session( "mothergoose_conn" )
24  Else ' If not open one
25      Set conn = Server.CreateObject( "ADODB.Connection" )
26      Call conn.Open( "mothergoose", "", "" )
27      Set Session( "mothergoose_conn" ) = conn
28  End If
29
30  ' Prepare the SQL statement that collects all
31  ' the information for this page

```

**1.1 Check for database connection; if none exists, open one**



```

32 sql = "SELECT product.title, language.language " & _
33       "FROM language " & _
34       "INNER JOIN (product INNER JOIN translations ON " & _
35       & "product.productID = translations.ProductID) ON " & _
36       & "language.languageid = translations.languageID " & _
37       & "ORDER BY translations.ProductID, " & _
38       " translations.languageID"
39 Set rs = Server.CreateObject( "ADODB.Recordset" )
40 Call rs.Open( sql, conn )
41
42 titleTest = "nothing"      ' Initialize titleTest
43 On Error Resume Next      ' Ignore errors
44 Call rs.MoveFirst()       ' Goto first record
45
46 ' Cycle through the records as long as there are more records
47 Do While Not rs.EOF
48     ' Check to see if we have printed this title before
49     If rs.Fields( "title" ).Value <> titleTest Then
50         ' If we have not seen the title before, print it out
51%>
52<P><FONT FACE = "arial" SIZE = "4">
53<% =rs.Fields( "title" ).Value %></FONT>
54<FONT FACE = "arial" SIZE = "1"><BR>
55Available in: </FONT>
56<%
57     End If
58
59     ' Check the value of the language field and
60     ' match it with the appropriate flag
61     Select Case rs.Fields( "language" ).Value
62         Case "English"

```

## 1.2 Create SQL query

## 1.3 Cycle through recordset

```
63         image = "/images/us.gif"
64     Case "French"
65         image = "/images/france.gif"
66     Case "Italian"
67         image = "/images/italy.gif"
68     Case "German"
69         image = "/images/germany.gif"
70     Case "Spanish"
71         image = "/images/spain.gif"
72 End Select
73
74 ' Now print the image using the value of the
75 ' language field to provide an alt tag value
76%>
77<IMG SRC = "<% =image %>"
78ALT = "<% =rs.Fields( "language" ).Value %>"
79HEIGHT = "25">
80<%
81     ' set titleTest to the current title
82     titleTest = rs.Fields( "title" ).Value
83
84     Call rs.MoveNext()     'move to the next record
85 Loop
86
87 ' Add in the footer and end the page
88%>
89<!--#include virtual = "/includes/mgtfooter.inc"--></P>
90</BODY>
91</HTML>
```

# Web browser showing the page translation.asp



## 11.10 Case Study: A Product Catalog

- Database:
  - `catalog.mdb`
  - Must be registered as DSN on server
- Use server-side includes for consistent look

```

1<% @LANGUAGE = VBScript %>
2<% Option Explicit %>
3
4<% ' Fig. 11.19 : catalog.asp %>
5
6<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.0 Transitional//EN">
7<HTML>
8<HEAD>
9<TITLE>Textbook Information Center: Technology</TITLE>
10</HEAD>
11
12<!-- #include virtual = "/includes/header.inc" -->
13
14<CENTER>
15<H2>
16<FONT COLOR = "brown">
17Welcome to the Textbook Information Center</FONT>
18</H2>
19</CENTER>
20
21<TABLE BORDER = 0 WIDTH = 100% CELLPADDING = 0 CELLSPACING = 0>
22<TR>
23<TD WIDTH = 25%>&nbsp;</TD>
24<TD BGCOLOR = "#008080" WIDTH = 10%>&nbsp;</TD>
25<TD BGCOLOR = "#008080" WIDTH = 40%>
26<FONT COLOR = "yellow" SIZE = 2>To view available titles
27for a technology<BR>
28select a link from this list:</FONT><P>
29
30<%
31  Dim dbConn, dbQuery, techRS
32

```

## 1.1 Include header with SSI statement

```

33 ' Check for an active connection, if not open one
34 If IsObject( Session( "catalog_dbConn" ) ) Then
35     Set dbConn = Session( "catalog_dbConn" )
36 Else
37     Set dbConn = Server.CreateObject( "ADODB.Connection" )
38     Call dbConn.Open( "catalog", "", "" )
39     Set Session( "catalog_dbConn" ) = dbConn
40 End If
41
42 ' Create the SQL query and create the recordset
43 dbQuery = "SELECT * FROM tech"
44 Set techRS = Server.CreateObject( "ADODB.Recordset" )
45 Call techRS.Open( dbQuery, dbConn )
46
47 On Error Resume Next      ' If an error occurs, ignore it
48
49 ' Move to the first record of the recordset
50 Call techRS.MoveFirst()
51
52 ' Begin a Do/While loop that will continue to
53 ' iterate as long as there are records
54 Do While Not techRS.EOF
55     ' Create a HREF that passes the techid number to the
56     ' next page and use the text value of technology to
57     ' label the link
58%>
59
60<LI><A HREF = "titles.asp?techid = <% =techRS( "techID" ) %>">
61<FONT COLOR = "yellow" SIZE = 2>
62<% =techRS( "technology" ) %>
63</FONT></A>
64<%

```

**1.2 Connect to database, if not already connected**

**1.3 Create SQL query and recordset**

**1.4 Create list of hyper-links corresponding to technologies in recordset**

```
65      ' Move to the next record in the recordset
66      Call techRS.MoveNext()
67      Loop
68 %>
69 </FONT><P>
70 </TD>
71 <TD WIDTH = 25%>&nbsp;</TD>
72 </TR>
73 </TABLE>
74 <!-- #include virtual = "/includes/footer.inc" -->
75 </HTML>
```

# Browser displaying catalog.asp

Textbook Information Center: Technology - Microsoft Internet Explorer

File Edit View Favorites Tools Help

Address <http://localhost/catalog/catalog.asp>

**DEITEL**  
& ASSOCIATES INC.

**JAVA™ TRAINING**  
Java™, C, C++, Visual Basic®, Object Technology and Internet Training  
Low Cost, High Quality On-Site and Public Seminars  
[deitel@deitel.com](mailto:deitel@deitel.com) 978.579.9911 (Fax) 978.579.9955  
490B Boston Post Road, Suite 200 Sudbury, MA 01776

**Welcome to the Textbook Information Center**

To view available titles for a technology select a link from this list:

- [C](#)
- [C++](#)
- [Visual Basic](#)
- [Java](#)

Send questions or comments about this site to [deitel@deitel.com](mailto:deitel@deitel.com)  
© Copyright 1995-1999 by Deitel & Associates, Inc. All Rights Reserved.  
Thursday, September 09, 1999 08:59

Local intranet



```

76<BODY BGCOLOR = "#ffffff">
77<TABLE BORDER = "0" WIDTH = "600">
78<TR>
79<TD><A HREF = "http://www.deitel.com"><IMG
80   SRC = "/images/logotiny.gif" BORDER = 0></A></TD>
81<TD BGCOLOR = "#008080">
82<P ALIGN = "center">
83<SPAN STYLE = "font-family: Arial, Helvetica, sans-serif">
84<IMG SRC = "/images/banneranim.gif"
85ALT = "banneranim.gif (185545 bytes)" WIDTH = "481"
86   HEIGHT = "28"><BR>
87<FONT COLOR = "#FFFF00">Java&#153;, C, C++, Visual
88Basic<SUP>&#174;</SUP>,
89Object Technology and Internet Training<BR>
90Low Cost, High Quality On-Site and Public Seminars<BR>
91<SMALL><A HREF = "mailto:deitel@deitel.com">deitel@deitel.com
92</A>
93</FONT>&nbsp;
94<FONT COLOR = "#80FFFF"><STRONG>978.579.9911</STRONG>&nbsp;
95(Fax) <STRONG>978.579.9955</STRONG><BR>
96<STRONG>490B Boston Post Road</STRONG>,
97<STRONG>Suite 200 Sudbury, MA 01776</STRONG></FONT>
98</SMALL></SPAN></P>
99</TD>
100</TR>
101<TR><TD COLSPAN = 2>&nbsp;</TD></TR>
102<TR>
103<TD COLSPAN = 2>
104<FONT FACE = "arial" SIZE = 2>

```

header.inc inserts  
Deitel image and  
training banner

```
105</FONT>
106</TD></TR>
107<TR><TD COLSPAN = 2>&nbsp;</TD></TR>
108<TR><TD COLSPAN = 2 ALIGN = right>
109
110<SPAN STYLE = "font-family: sans-serif, Helvetica, Arial;
111font-size: 8pt">
112Send questions or comments about this site to
113<A HREF = "mailto:deitel@deitel.com">deitel@deitel.com</A><BR>
114&copy; Copyright 1995-1999 by Deitel & Associates, Inc.
115All Rights Reserved.<BR>
116Thursday, September 09, 1999 08:59<BR>
117</SPAN>
118</TD></TR>
119</TABLE>
120</BODY>
121</HTML>
```

**footer.inc inserts  
copyright information**

```

122<% @LANGUAGE = VBScript %>
123<% Option Explicit %>
124
125<% ' Fig. 11.22 : titles.asp %>
126
127<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.0 Transitional//EN">
128<HTML>
129<HEAD>
130<TITLE>Textbook Information Center: Titles</TITLE>
131</HEAD>
132<!-- #include virtual = "/includes/header.inc" -->
133
134<CENTER>
135<H2><FONT COLOR = "brown">Welcome to the Textbook Information
136Center</FONT></H2>
137</CENTER>
138<TABLE BORDER = 0 WIDTH = 100% CELLPADDING = 0 CELLSPACING = 0>
139<TR>
140<TD WIDTH = 15%>&nbsp;</TD>
141<TD BGCOLOR = "#008080" WIDTH = 5%>&nbsp;</TD>
142<TD BGCOLOR = "#008080" WIDTH = 60%>
143<FONT COLOR ="yellow" SIZE = 2>Select a title from the list below:
144</FONT><P>
145<UL>
146<%
147   Dim dbConn, dbQuery, titleRS
148
149   ' Check for an active connection, if not open one
150   If IsObject( Session( "catalog_dbConn" ) ) Then
151       Set dbConn = Session( "catalog_dbConn" )
152   Else

```

**1.1 Check for connection to database; if there is none, open one**

```

153     Set dbConn = Server.CreateObject( "ADODB.Connection" )
154     Call dbConn.Open( "catalog", "", "" )
155     Set Session( "catalog_dbConn" ) = dbConn
156 End If
157
158 ' Create the SQL query and create the recordset
159 dbQuery = "SELECT * FROM products WHERE techid="
160 dbQuery = dbQuery & Request( "techid" )
161 Set titleRS = Server.CreateObject( "ADODB.Recordset" )
162 Call titleRS.Open( dbQuery, dbConn )
163
164 On Error Resume Next ' If an error occurs, ignore it
165
166 ' Move to the first record of the recordset
167 Call titleRS.MoveFirst()
168
169 ' Begin a do while loop that will continue to
170 ' iterate as long as there are records
171 Do While Not titleRS.EOF
172
173     ' Create a HREF that passes the productid number to the
174     ' next page and use the title and the edition to label
175     ' the link
176%>
177
178<LI><A HREF = "description.asp?productid=<%
179=titleRS( "productID" ) %>">
180<FONT COLOR = "yellow" SIZE = 2>
181<% =titleRS( "title" ) %>
182</FONT></A>
183<FONT COLOR = "yellow" SIZE = 2> &nbsp; Edition
184<% =titleRS( "edition" ) %>

```

**1.2 Create SQL query and recordset**

**1.3 Cycle through recordset and create hyperlink for each product**

```
185</FONT>
186
187<%
188     Call titleRS.MoveNext()    ' Move to the next record
189 Loop
190%>
191
192</UL>
193</FONT><P>
194</TD>
195
196<TD BGCOLOR = "#008080" WIDTH = 5%>&nbsp;&nbsp;&nbsp;</TD>
197
198<TD WIDTH = 15%></TD>
199</TR>
200</TABLE>
201
202<% ' Provide a link back to the previous page %>
203<P ALIGN = CENTER>
204Back to the <STRONG>
205<A HREF = "catalog.asp">Technology List</A><STRONG></P>
206
207<!-- #include virtual = "/includes/footer.inc" -->
208</HTML>
```

# Browser displaying titles.asp



The screenshot shows a Microsoft Internet Explorer browser window. The title bar reads 'Textbook Information Center: Titles - Microsoft Internet Explorer'. The address bar contains 'http://localhost/catalog/titles.asp?techid=4'. The main content area features the Deitel & Associates Inc. logo on the left and a green banner for 'JAVA™ TRAINING' on the right. Below the banner, a red heading reads 'Welcome to the Textbook Information Center'. A green box contains the text 'Select a title from the list below:' followed by a bulleted list of two items: 'Java : How to Program, Edition 2' and 'Java : How to Program, Edition 3'. At the bottom, there is a link 'Back to the Technology List' and a footer with contact information and a copyright notice.

Textbook Information Center: Titles - Microsoft Internet Explorer

File Edit View Favorites Tools Help

Address <http://localhost/catalog/titles.asp?techid=4>

**DEITEL**  
& ASSOCIATES INC.

**JAVA™ TRAINING**  
Java™, C, C++, Visual Basic®, Object Technology and Internet Training  
Low Cost, High Quality On-Site and Public Seminars  
[deitel@deitel.com](mailto:deitel@deitel.com) 978.579.9911 (Fax) 978.579.9955  
490B Boston Post Road, Suite 200 Sudbury, MA 01776

**Welcome to the Textbook Information Center**

Select a title from the list below:

- [Java : How to Program, Edition 2](#)
- [Java : How to Program, Edition 3](#)

Back to the [Technology List](#)

Send questions or comments about this site to [deitel@deitel.com](mailto:deitel@deitel.com)  
© Copyright 1995-1999 by Deitel & Associates, Inc. All Rights Reserved.  
Thursday, September 09, 1999 08:59

Local intranet

```

209<% @LANGUAGE = VBScript %>
210<% Option Explicit %>
211
212<% ' Fig. 11.23 : description.asp %>
213
214<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.0 Transitional//EN">
215<HTML>
216<HEAD>
217<TITLE>Textbook Information Center: Description
218</TITLE>
219</HEAD>
220<!-- #include virtual = "/includes/header.inc" -->
221
222<%
223     Dim dbConn, dbQuery, productRS, isbnString
224
225     ' Check for an active connection, if not open one
226     If IsObject( Session( "catalog_dbConn" ) ) Then
227         Set dbConn = Session( "catalog_dbConn" )
228     Else
229         Set dbConn = Server.CreateObject( "ADODB.Connection" )
230         Call dbConn.Open( "catalog", "", "" )
231         Set Session( "catalog_dbConn" ) = dbConn
232     End If
233
234     ' Create the SQL query and create the recordset
235     dbQuery = "SELECT * FROM products WHERE productid="
236     dbQuery = dbQuery & Request( "productid" )
237     Set productRS = Server.CreateObject( "ADODB.Recordset" )
238     Call productRS.Open( dbQuery, dbConn )
239%>
240

```

**1.1 Check for database connection**

**1.2 Create SQL query and recordset**





```
273
274   ' Use the isbn from the database to create a link directly
275   ' to the book at amazon.com
276%>
277<A HREF = "http://www.amazon.com/exec/obidos/ASIN/<%
278   =isbnString %>/deitelassociatin">
279<IMG SRC = "/images/amazonw.gif" ALT = "amazonw.gif (1557 bytes)"
280     WIDTH = "90" HEIGHT = "29"></A>
281</P>
282</FONT><P>
283</TD>
284
285<TD WIDTH = 25%>&nbsp;</TD>
286
287</TR>
288</TABLE>
289
290<P ALIGN = CENTER>Back to the <STRONG>
291<A HREF = "titles.asp?techid=<% =productRS( "techID" ) %>">Titles
292</A></STRONG>
293
294<!-- #include virtual = "/includes/footer.inc" -->
295</HTML>
```

## 1.4 Provide link to Amazon.com

